

Postestimation commands

Remarks and examples

Also see

predict

Methods and formulas

margins

Reference

Postestimation commands

The following postestimation commands are available after `reg3`:

Command	Description
<code>contrast</code>	contrasts and ANOVA-style joint tests of parameters
* <code>estat ic</code>	Akaike's, consistent Akaike's, corrected Akaike's, and Schwarz's Bayesian information criteria (AIC, CAIC, AICc, and BIC, respectively)
<code>estat summarize</code>	summary statistics for the estimation sample
<code>estat vce</code>	variance–covariance matrix of the estimators (VCE)
<code>estimates</code>	cataloging estimation results
<code>etable</code>	table of estimation results
<code>forecast</code>	dynamic forecasts and simulations
<code>hausman</code>	Hausman's specification test
<code>lincom</code>	point estimates, standard errors, testing, and inference for linear combinations of parameters
<code>margins</code>	marginal means, predictive margins, marginal effects, and average marginal effects
<code>marginsplot</code>	graph the results from margins (profile plots, interaction plots, etc.)
<code>nlcom</code>	point estimates, standard errors, testing, and inference for nonlinear combinations of parameters
<code>predict</code>	predictions and their SEs, residuals, etc.
<code>predictnl</code>	point estimates, standard errors, testing, and inference for generalized predictions
<code>pwcompare</code>	pairwise comparisons of parameters
<code>test</code>	Wald tests of simple and composite linear hypotheses
<code>testnl</code>	Wald tests of nonlinear hypotheses

*`estat ic` is not appropriate after `reg3`, `2sls`.

predict

Description for predict

predict creates a new variable containing predictions such as linear predictions, standard errors, residuals, and differences between the linear predictions of two equations.

Menu for predict

Statistics > Postestimation

Syntax for predict

predict [<i>type</i>] <i>newvar</i> [<i>if</i>] [<i>in</i>] [, <u>e</u> quation(<i>eqno</i> [, <i>eqno</i>]) <i>statistic</i>]	
<i>statistic</i>	Description

Main	
<u>x</u> b	linear prediction; the default
<u>s</u> tdp	standard error of the linear prediction
<u>r</u> esiduals	residuals
<u>d</u> ifference	difference between the linear predictions of two equations
<u>s</u> tddp	standard error of the difference in linear predictions

These statistics are available both in and out of sample; type predict ... if e(sample) ... if wanted only for the estimation sample.

Options for predict

Main

equation(*eqno*[, *eqno*]) specifies to which equation you are referring.

equation() is filled in with one *eqno* for the *xb*, *stdp*, and *residuals* options. equation(#1) would mean the calculation is to be made for the first equation, equation(#2) would mean the second, and so on. You could also refer to the equations by their names. equation(income) would refer to the equation named income and equation(hours) to the equation named hours.

If you do not specify equation(), results are the same as if you specified equation(#1).

difference and stddp refer to between-equation concepts. To use these options, you must specify two equations, for example, equation(#1,#2) or equation(income,hours). When two equations must be specified, equation() is required.

xb, the default, calculates the linear prediction (fitted values)—the prediction of $\mathbf{x}_j\mathbf{b}$ for the specified equation.

stdp calculates the standard error of the prediction for the specified equation. It can be thought of as the standard error of the predicted expected value or mean for the observation's covariate pattern. The standard error of the prediction is also referred to as the standard error of the fitted value.

residuals calculates the residuals.

difference calculates the difference between the linear predictions of two equations in the system. With `equation(#1,#2)`, difference computes the prediction of equation (#1) minus the prediction of equation (#2).

stdp is allowed only after you have previously fit a multiple-equation model. The standard error of the difference in linear predictions ($\mathbf{x}_{1j}\mathbf{b} - \mathbf{x}_{2j}\mathbf{b}$) between equations 1 and 2 is calculated.

For more information on using predict after multiple-equation estimation commands, see [\[R\] predict](#).

margins

Description for margins

margins estimates margins of response for linear predictions and differences between the linear predictions of two equations.

Menu for margins

Statistics > Postestimation

Syntax for margins

```
margins [marginlist] [, options]  
margins [marginlist] , predict(statistic ...) [predict(statistic ...) ...] [options]
```

<i>statistic</i>	Description
default	linear predictions for each equation
xb	linear prediction for a specified equation
difference	difference between the linear predictions of two equations
stdp	not allowed with margins
residuals	not allowed with margins
stddp	not allowed with margins

xb defaults to the first equation.

Statistics not allowed with margins are functions of stochastic quantities other than $e(b)$.


For the full syntax, see [\[R\] margins](#).

Remarks and examples

► Example 1: Using predict

In [example 2](#) of [\[R\] reg3](#), we fit a simple supply-and-demand model. Here we obtain the fitted supply and demand curves assuming that the exogenous regressors equal their sample means. We first replace each of the three exogenous regressors with their sample means, then we call `predict` to obtain the predictions.

```
. use https://www.stata-press.com/data/r19/supDem
. global demand "(qDemand: quantity price pcompete income)"
. global supply "(qSupply: quantity price praw)"
. reg3 $demand $supply, endog(price)
(output omitted)
. summarize pcompete, meanonly
. replace pcompete = r(mean)
(49 real changes made)
. summarize income, meanonly
. replace income = r(mean)
(49 real changes made)
. summarize praw, meanonly
. replace praw = r(mean)
(49 real changes made)
. predict demand, equation(qDemand)
(option xb assumed; fitted values)
. predict supply, equation(qSupply)
(option xb assumed; fitted values)
. graph twoway line demand price, sort || line supply price, ytitle(" ")
> legend(label(1 "Fitted values: qDemand") label(2 "Fitted values: qSupply"))
```



As we would expect based on economic theory, the demand curve slopes downward while the supply curve slopes upward. With the exogenous variables at their mean levels, the equilibrium price and quantity are slightly less than 33 and 13, respectively.

► Example 2: Obtaining forecasts

In [example 3](#) of [\[R\] reg3](#), we fit Klein's (1950) model of the US economy. That model includes three stochastic equations we fit using `reg3` as well as four identities. Here we briefly illustrate how the `forecast` command can be used to obtain forecasts for all the endogenous variables in the model. For a more detailed discussion of how to forecast with this model, see [\[TS\] forecast](#).


In Stata, we type

```
. use https://www.stata-press.com/data/r19/klein2, clear
. reg3 (c p L.p w) (i p L.p L.k) (wp y L.y yr), endog(w p y) exog(t wg g)
  (output omitted)
. estimates store kleineqs
. forecast create kleinmodel
  Forecast model kleinmodel started.
. forecast estimates kleineqs
  Added estimation results from reg3.
  Forecast model kleinmodel now contains 3 endogenous variables.
. forecast identity y = c + i + g
  Forecast model kleinmodel now contains 4 endogenous variables.
. forecast identity p = y - t - wp
  Forecast model kleinmodel now contains 5 endogenous variables.
. forecast identity k = L.k + i
  Forecast model kleinmodel now contains 6 endogenous variables.
. forecast identity w = wg + wp
  Forecast model kleinmodel now contains 7 endogenous variables.
. forecast solve, begin(1937)
Computing dynamic forecasts for model kleinmodel.
```

```
Starting period: 1937
Ending period:  1941
Forecast prefix: f_
1937: .....
1938: .....
1939: .....
1940: .....
1941: .....
Forecast 7 variables spanning 5 periods.
```

Here we have obtained dynamic forecasts for our 7 endogenous variables beginning in 1937. By default, the variables containing the forecasts begin with the prefix `f_`. Next we plot the forecast and actual values of consumption:

```
. tsline c f_c
```



For more information about producing forecasts, see [\[TS\] forecast](#).



Methods and formulas

The computational formulas for the statistics produced by `predict` can be found in [\[R\] predict](#) and [\[R\] regress postestimation](#).

Reference

Klein, L. R. 1950. *Economic Fluctuations in the United States 1921–1941*. New York: Wiley.

Also see

[\[R\] reg3](#) — Three-stage estimation for systems of simultaneous equations

[\[U\] 20 Estimation and postestimation commands](#)

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).